Artin L-functions

B. Allombert

IMB
CNRS/Université de Bordeaux

28/03/2019

This project has received funding from the European Union's Horizon 2020 research and innovation programme under grant agreement N° 676541

Hecke L functions

We compute the Hecke L-function attached to a non-trivial character of the class group of the ring of integers of $\mathbb{Q}(\sqrt{-23})$.

```
? bnf = bnfinit(a^2+23);
? bnr = bnrinit(bnf, 1);
? bnr.clgp
%3 = [3,[3]]
? Hecke = lfuncreate([bnr,[1]]); Hecke[2..5]
%4 = [1,[0,1],1,23]
```

Hecke L functions

We compute $L(0)$ and $L'(0)$.

```
? lfun(Hecke,0)
%5 = 0
? z=lfun(Hecke,0,1)
%6 = 0.28119957432296184651205076406787829979+0.E-6
? P=algdep(exp(z),3)
%7 = x^3-x-1
? nfdisc(P)
%8 = -23
```

Hecke L functions

The L function should be attached to a modular form of level 23 and weight 1 with character -23.

```
? mf = mfinit([23,1,-23],0);
? S=mfeigenbasis(mf); #S
%10 = 1
? round(mfcoefs(S[1],100)[^1]-lfunan(Hecke,100),&e)
%11 = [0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,
? e
%12 = -127
```

Let E be the elliptic curve with Cremona label "40a1".
$E : y^2 + y = x^3 - 7x - 6$, we build the field $\mathbb{Q}(E[3])$ generated by the coordinates of the points of 3-torsions.

```
? E=ellinit("40a1");
? P=elldivpol(E,3)
%2 = 3*x^4-42*x^2-72*x-49
? Q=polresultant(P,y^2-elldivpol(E,2));
%3 = 27*y^8-5184*y^6-115200*y^4-40960000
? polgalois(Q)
%4 = [48,-1,23,"2S_4(8)=GL(2,3)"]
? R=nfsplitting(Q)
%5 = y^48-48*y^46+1128*y^44-15792*y^42+147024*y^40-
```

This defines a Galois extension of $\mathbb{Q}$ with Galois group $\mathrm{GL}_2(\mathbb{F}_3)$.

Non monomial representation

```
? N=nfinit(R); G=galoisinit(N);
? [T,o]=galoischartable(G); T~
%6 =
%   [1  1        1  1  1  1  1        1]
%   [1 -1       -1  1  1  1  1       -1]
%   [2  0        0  2 -1 -1  2        0]
%   [2  0  -y^3-y -2 -1  1  0    y^3+y]
%   [2  0    y^3+y -2 -1  1  0   -y^3-y]
%   [3 -1        1  3  0  0 -1        1]
%   [3  1       -1  3  0  0 -1       -1]
%   [4  0        0 -4  1 -1  0        0]
? o
%7 = 8
```

$o = 8$ means that the variable y denotes a 8-th root of unity.

Non monomial representation

```
? minpoly(Mod(y^3+y, polcyclo(o,y)))
%8 = x^2+2
```

So the coefficients are in $\mathbb{Q}(\sqrt{-2})$. We use the fourth irreducible representation.

```
? L = lfunartin(N,G,T[,4],o);
? L[2..5]
%10 = [1,[0,1],1,1800]
? lfuncheckfeq(L)
%11 = -125
```

Determinant

```
? dT = galoischardet(G,T[,4],o)
%12 = [1,-1,-1,1,1,1,1,-1]~
? dL = lfunartin(N,G,dT,o); dL[2..5]
%13 = [0,[1],1,3];
```

So L is associate to a modular form of weight 1, level 1800 and Nebentypus $\left(\frac{-3}{\cdot}\right)$.

Link to E

We reduce the coefficients of L modulo $1-\sqrt{-2}$ of norm 3.

```
? S = lfunan(L,1000); SE = lfunan(E,1000);
? Smod3 = round(real(S))+round(imag(S)/sqrt(2));
? [(Smod3[i]-SE[i])%3|i<-[1..#Smod3],gcd(i,40)==1]
%16 = [0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,...
```

The coefficients of L are congruent to the coefficients of the L-function associated to E modulo $1-\sqrt{-2}$.

Modular forms

Both E and L are attached to modular forms.

```
? mfE=mfinit([40,2,1],0);
? eiE = mfeigenbasis(mfE); #eiE
%18 = 1
? mfcoefs(eiE[1],100)[^1]==ellan(E,100)
%19 = 1
```

Modular forms

```
? mfL=mfinit([1800,1,-3],0);
? eiL = mfeigenbasis(mfL); #eiL
%20 = 2
? round(subst(lift(mfcoefs(eiL[2],100)),y,sqrt(-2))
%22 = [0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,
? e
%23 = -127
```

Modular forms

```
? mfgaloistype(mfL,eiL[2])
%24 = -24
? PR=polredabs(mfgaloisprojrep(mfL,eiL[2]))
%25 = x^24-12*x^23+70*x^22-264*x^21+727*x^20-1572*x
? PR==polredabs(nfsplitting(elldivpol(E,3)))
%26 = 1
```